

	
3GPP TSG-RAN5 Meeting #2021-TTCN email	R5s211173
Online, , 7th Dec 2020 - 31st Dec 2021
	CR-Form-v12.1

	CHANGE REQUEST

	

	
	38.523-3
	CR
	1772
	rev
	1
	Current version:
	16.2.0
	

	

	For HELP on using this form: comprehensive instructions can be found at 
http://www.3gpp.org/Change-Requests.

	



	Proposed change affects:
	UICC apps
	
	ME
	
	Radio Access Network
	
	Core Network
	



	

	Title:	
	Addition of NR5GC test case 11.3.6 in FR1

	
	

	Source to WG:
	Datang LinkTester Technology

	Source to TSG:
	R5

	
	

	Work item code:
	5GS_NR_LTE-UEConTest
	
	Date:
	2021-06-21

	
	
	
	
	

	Category:
	B
	
	Release:
	Rel-16

	
	Use one of the following categories:
F  (correction)
A  (mirror corresponding to a change in an earlier 													release)
B  (addition of feature), 
C  (functional modification of feature)
D  (editorial modification)
Detailed explanations of the above categories can
be found in 3GPP TR 21.900.
	Use one of the following releases:
Rel-8	(Release 8)
Rel-9	(Release 9)
Rel-10	(Release 10)
Rel-11	(Release 11)
…
Rel-15	(Release 15)
Rel-16	(Release 16)
Rel-17	(Release 17)
Rel-18	(Release 18)

	
	

	Reason for change:
	To add Rel-15 NR5GC test case 11.3.6 to the NR5GC ATS

	
	

	Summary of change:
	This document lists all changes applied to NR5GC test case  11.3.6　required for approval. See detailed change description for further information.

	
	

	Consequences if not approved:
	Test case will not be added to ATS.

	
	

	Clauses affected:
	11.3.6.NR5GC

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications	
	TS/TR ... CR ... 

	affected:
	X
	
	 Test specifications
	38.523-1

	(show related CRs)
	
	X
	 O&M Specifications
	TS/TR ... CR ... 

	
	

	Other comments:
	

	
	

	This CR's revision history:
	



Page 1


[bookmark: _Toc75197275]Table of Contents
Table of Contents	2
1	Overview	3
1.1	Verification Test Summary	3
2	Corrections required	3
2.1	Change 1	3
2.2	Change 2	5
3	Branches executed	13
4	Execution Log Files	14
4.1	HiSilicon Balong 5000	14
5	References	14


1. [bookmark: _Toc122434485][bookmark: _Toc75197276]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.3.6 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Pengchong Shen
	shenpengchong@datangmobile.cn
0. [bookmark: _Toc75197277]Verification Test Summary
Test Case: 	11.3.6
ATS Version:	iwd-TTCN3-B2020-09_D21wk12
System Simulator used:	Datang LinkTester 5G Protocol Conformance Test System ECT9610
UE used:	HiSilicon Balong 5000
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc75197278]Corrections required
1. [bookmark: _Toc75197279][bookmark: OLE_LINK4][bookmark: OLE_LINK5][bookmark: OLE_LINK6]Change 1
	Function name
	f_TC_11_3_6_NR5GC

	Reason for change
	1. To make sure the procedure of moving UE into inactive state completely befor UE loop back the IP PDU received at step2, it shall set the ip delay to 2 seconds.
2. According to passed CR R5-213506, uac-BarringForAccessIdentity should be modify.
3. The current TTCN is missed postamble for pc_inactiveState set to false.
4. According to passed CR R5-213506, UE should finish the postamble at RRC connected state.

Note： it is supported with a prose CR R5-213506 and a draft prose CR.

	Summary of change
	1. Modify the '01'O to '02'O.
2. Modify  '1111111'B to  '1011111'B.
3. Add procedure when pc_inactiveState is set to false during postamble.
4. Modify STATE_IDLE_1A to STATE_CONNECTED_3A.

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	



Before Change
	function f_TC_11_3_6_NR5GC() runs on NR5GC_PTC
  { // UAC / Access Identity 2 / New cell not in the country of its HPLMN/EHPLMN  0% access probability/MCS indicator / HPLMN/0%/100% accessibility AC7/RRC_INACTIVE

    var Octet_Type v_IP_PDU_delay := '01'O;

    f_NR5GC_Init(NR_1);

    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(3,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));

    //Set maximum cell power level for the Cells to be used
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -80, -80);  // Max Power level for FR2 is FFS
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell12, -80, -80);  // Max Power level for FR2 is FFS

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(nr_Cell12);

    //The UE is equipped with a USIM configuration as defined in TS 38.508-1 [4] Table 6.4.1-19
    f_UT_USIM_Insert (UT, "USIM Configuration 19 specified in Table 6.4.1-19 TS 38.508-1");

    f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A, TEST_LOOPModeB_ON, -, -, v_IP_PDU_delay);

    f_NR_TestBody_Set(true);

    fl_TC_11_3_6_TestBody();

    f_NR_TestBody_Set(false);

    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);
    f_NR_Postamble(nr_Cell1, STATE_IDLE_1A);
  }



After Change
	  function f_TC_11_3_6_NR5GC() runs on NR5GC_PTC
  { // UAC / Access Identity 2 / New cell not in the country of its HPLMN/EHPLMN  0% access probability/MCS indicator / HPLMN/0%/100% accessibility AC7/RRC_INACTIVE

    var Octet_Type v_IP_PDU_delay := '02'O;

    f_NR5GC_Init(NR_1);

    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(3,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1011111'B)));

    //Set maximum cell power level for the Cells to be used
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell1, -80, -80);  // Max Power level for FR2 is FFS
    f_NR_CellInfo_InitMaxReferencePower(nr_Cell12, -80, -80);  // Max Power level for FR2 is FFS

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(nr_Cell12);

    //The UE is equipped with a USIM configuration as defined in TS 38.508-1 [4] Table 6.4.1-19
    f_UT_USIM_Insert (UT, "USIM Configuration 19 specified in Table 6.4.1-19 TS 38.508-1");

    f_NR5GC_Preamble(nr_Cell1, STATE_IDLE_1A, TEST_LOOPModeB_ON, -, -, v_IP_PDU_delay);

    f_NR_TestBody_Set(true);

    fl_TC_11_3_6_TestBody();

    f_NR_TestBody_Set(false);

	if(pc_inactiveState){
		
	 f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1);
		
	 f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);

	}else{

    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell12);
    f_NR_Postamble(nr_Cell12, STATE_CONNECTED_3A);
   }
 }


1. [bookmark: _Toc75197280][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 2
	Function name
	fl_TC_11_3_6_TestBody

	Reason for change
	1. CLOSE UE TEST LOOP procedure is missing for this TC.
2. USIM configuration 19 configured Access Identity 2 is used for this case.  The service type IE in the SERVICE REQUEST message shall be set to "high priority access"  at step 11 according to TS 24.501 clause 5.6.1.2.1
3. According to passed CR R5-213506, 10 s is not enough to find and camp on a new NR cell for UE at step 2.
4. If step 13 RRC connection is released, UE will not receive Testmode and Closeloop deactivation commands from N/W as part of the TTCN Postamble when pc_inactiveState is set false.
5. According to passed CR R5-213506, add step 14a14A1 - 14a14A6.
6. According to passed CR R5-213506, voided Step 14a19 as in RRC_IDLE mode, UE will not receive Testmode and Closeloop deactivation commands from N/W as part of the TTCN Postamble.
Note: point 1and 2 is supported with a draft prose CR.

	A
	1. Add step 6E and 6F.
2. Modify the '0001'B to '0101'B.
3. According to passed CR R5-213506 modfiy the procedure of step 1A and modify the timer 10.0 to 20.0 for step 1.
4. Perform step 13 when the pc_inactiveState is set to ture.
5. Add procedure for steps 14a14A1 - 14a14A6.
6. Delete step 14a19.

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	


Before change
	  function fl_TC_11_3_6_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters(nr_Cell1);
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var integer v_DRBId;
    var template (value) QosFlow_Identification_Type v_QoS_Id;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T2;
    var float v_TimerValue1 := 10.0;
    var float v_TimerValue2 := 30.0; //@sic R5-201227 sic@
    timer t_WaitTime;

    v_CellPowerList_T1 := {
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, -80, -80)  // Power level for FR2 is FFS
    };

    v_CellPowerList_T2 := {
      cs_NR_CellPower (nr_Cell1, -80, -80),  // Power level for FR2 is FFS
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };

    //@siclog "Step 1" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.6.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_T1);

    //@siclog "Step 2" siclog@
    //The UE transmit a RRCSetupRequest message within 10 s on NR cell 12.
    //Check: Does the RRCSetupRequest message includ the establishmentCause which is any value except mcs-PriorityAccess?
    t_WaitTime.start(v_TimerValue1);
    alt {
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest())) -> value v_ReceivedAsp
        {
          t_WaitTime.stop;
          if(match(v_ReceivedAsp.Signalling.Rrc.Ccch.message_.c1.rrcSetupRequest.rrcSetupRequest.establishmentCause, mcs_PriorityAccess))
          {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2");
          }
          else
          {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
          }
        }
      [] t_WaitTime.timeout{}
    }

    //@siclog "Steps 3 - 6" siclog@
    //Steps 2 to 5 of the mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 12. (Note 1)
    //Note 1: The SS includes a 5GS network feature support IE in the REGISTRATION ACCEPT message configured as Table 11.3.6.3.3-3.
    f_NR_RRCSetup_Def(nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def));
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }

    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease, cs_NG_NetworkFeatureSupport_TC1136);

    //@siclog "Step 6A - 6D" siclog@
    //@sic R5-201227 sic@
    //Steps 5-8 as defined in TS 38.508-1 [4] Table Table 4.5.4.2-3.
    f_NR_RRC_ActivateSecurity (nr_Cell12, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
    f_NR_RRCReconfigExistingDRBs (nr_Cell12, omit);

    //@siclog "Step 7" siclog@
    //The SS transmits one IP PDU.
    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    v_QoS_Id := f_NR_GetQosFlowIdentification_ForDRBId(v_DRBId);
    DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell12, v_QoS_Id, {v_IPData}));

    //@siclog "Step 8" siclog@
    //The SS transmits an RRCRelease message and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell12);

    //@siclog "Step 9" siclog@
    //Check: Does the UE transmit a RRCSetupRequest message including establishmentCause of mcs-PriorityAccess within 10s?
    t_WaitTime.start(v_TimerValue1);
    f_NR_RRC_SetupRequest_Common(nr_Cell12, t_WaitTime, mcs_PriorityAccess);  //@sic R5s201387 sic@

    //@siclog "Step 10" siclog@
    //SS transmit an RRCSetup message.
    f_NR_RRCSetup_Def(nr_Cell12);

    //@siclog "Step 11" siclog@
    //The UE transmits an RRCSetupComplete message to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), v_S_TMSI, '0001'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }

    //@siclog "Step 11A - 11D" siclog@
    //@sic R5-201227 sic@
    //Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Step 12" siclog@
    //The UE loops back the IP PDU.
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell12, v_QoS_Id, {v_IPData}));

    //@siclog "Step 13" siclog@
    //The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell12);

    //EXCEPTION: Steps 14a1-14a19 describe behaviour that depends on UE configuration; the "lower case letter" identifies a step sequence that takes place if inactiveState is configured
    //@sic R5-211409 sic@
    if(pc_inactiveState){

      //@siclog "Step 14a1" siclog@
      //IF pc_inactiveState THEN the SS changes the SIB1 of NR Cell 1 to set the uac-BarringInfo and adjusts the NR Cells power levels according to row "T2" in table 11.3.6.3.2-1/2.
      f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                           cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(7,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));
      f_NR_ModifySysinfo(nr_Cell1, false); //Modify Sysinfo in SS
      
      f_NR_SetCellPowerList(v_CellPowerList_T2);
      
      //@siclog "Steps 14a2 - 14a6" siclog@
      //Steps 1 to 5 of the mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 1. (Note 2)
      //Note 2: The UE performs registration and the RRC connection is released.
      v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell1, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected); // @sic R5-200477 sic@
      if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
      }
      f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1,v_GMM_MobilityInfo, omit, omit, noRrcConnectionRelease);
      
      //@siclog "Step 14a7 - 14a10" siclog@
      //@sic R5-201227 sic@
      //Steps 5-8 as defined in TS 38.508-1 [4] Table Table 4.5.4.2-3.
      f_NR_RRC_ActivateSecurity (nr_Cell1, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
      f_NR_RRCReconfigExistingDRBs (nr_Cell1, omit);
      
      //@siclog "Step 14a11" siclog@
      //The SS transmits one IP PDU.
      DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id, {v_IPData}));
      
      //@siclog "Step 14a12" siclog@
      //The SS transmits an RRCRelease message with suspend configuration and move the UE to RRC_INACTIVE.
      f_NR_RRC_InactiveDef(nr_Cell1);
      
      //@siclog "Step 14a13 siclog@
      //Check: Does the UE transmit a RRCResumeRequest message including resumeCause of mcs-PriorityAccess within 10s?
      //Note 4: The wait time 10s is selected to be less than T390 minimum = (0.7 + 0.6 * rand) * uac-BarringTime(16s) = 11.2s when 'rand' takes the minimum value of 0.
      t_WaitTime.start(v_TimerValue1);
      alt {
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mcs_PriorityAccess)))
        {
           t_WaitTime.stop;
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22");
        }
        [] t_WaitTime.timeout {}
      }
      
      //@siclog "Step 14a14 siclog@
      //SS changes SIB1 according to Table 11.3.6.3.3-5. And the SS notifies the UE of change of System Information on NR Cell 1 by send Short Message on PDCCH using P-RNTI.
      f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1, omit);
      f_NR_ModifySysinfo(nr_Cell1, true, RRC_INACTIVE);
      
      //@siclog "Step 14a15 siclog@
      //Check: Does the UE transmit a RRCResumeRequest message including resumeCause of mcs-PriorityAccess within 30 s?(Note 3)
      //Note 3: The wait time 30s is selected to cover (0.7 + 0.6 * rand) * uac-BarringTime(16s) = 20.8s + 1s (IP PDU delay timer) + 5.12s (modification period) = 26.92s rounded up to 27s when 'rand' takes the maximum value of 1.
      t_WaitTime.start(v_TimerValue2);
      alt {
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mcs_PriorityAccess)))
        {
           t_WaitTime.stop;
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 24");
        }
        [] t_WaitTime.timeout {}
      }
      
      //@siclog "Step 14a16 siclog@
      //SS transmit an RRCResume message.
      f_NR_SendRRCResume_Def(nr_Cell1);
      
      //@siclog "Step 14a17" siclog@
      //The UE transmits an RRCResumeComplete message to confirm the successful completion of the connection establishment.
      f_NR_RRCResumeComplete_Def(nr_Cell1);
      
      //@siclog "Step 14a18 siclog@
      //The UE loop back the IP PDU.
      DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id, {v_IPData}));
      
      //@siclog "Step 14a19 siclog@
      //The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
      f_NR_RRCRelease(nr_Cell1);
    }

    //@siclog "Steps 15 - 28 siclog@
    //@sic R5-211409 sic@
    //Void
  }


After  change
	  function fl_TC_11_3_6_TestBody() runs on NR5GC_PTC
  {
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters(nr_Cell1);
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var octetstring v_IPData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());
    var integer v_DRBId;
    var template (value) QosFlow_Identification_Type v_QoS_Id;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T2;
    var float v_TimerValue1 := 20.0;
    var float v_TimerValue2 := 30.0; //@sic R5-201227 sic@
    timer t_WaitTime;
	
	var PLMN_Identity v_ASN_PLMN := f_NR_CellInfo_GetPLMN (nr_Cell12);
	var NAS_PlmnId v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);
	var template (omit) NG_NAS_MSG_Indication_Type v_ReceivedMsgTemplate;
	var template(omit) RadioBearerConfig v_RadioBearerConfig;
	var template (value) NR_RLC_BearerToAddModList_Type v_NR_RLC_BearerToAddModList;
    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP (nr_Cell1, tsc_NR_BWP_Id);
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template(value) SpCellConfig v_SpCellConfig;	

    v_CellPowerList_T1 := {
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, -80, -80)  // Power level for FR2 is FFS
    };

    v_CellPowerList_T2 := {
      cs_NR_CellPower (nr_Cell1, -80, -80),  // Power level for FR2 is FFS
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };

    //@siclog "Step 1" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.6.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_T1);

	//@siclog "Step 1A" siclog@
	//Check: Does the UE transmit a RRCSetupRequest message within 20 s on NR cell 12? (Note 4)	
	t_WaitTime.start(v_TimerValue1);
		alt {
		  [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest())) -> value v_ReceivedAsp
			{
			  t_WaitTime.stop;
			  f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 1A");
			}
		  [] t_WaitTime.timeout{f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 1A");}
		}
 	//@siclog "Step 1B" siclog@
	//SS changes SIB1 of NR Cell 12 according to 38.508-1 [4] Table 4.6.1-28 and sends Short Message on PDCCH using P-RNTI.	
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell12, omit);	
	f_NR_ModifySysinfo(nr_Cell12);
	
    //@siclog "Step 2" siclog@
    //The UE transmit a RRCSetupRequest message within 10 s on NR cell 12.
    //Check: Does the RRCSetupRequest message includ the establishmentCause which is any value except mcs-PriorityAccess?
   
       SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell12, cr_38508_RRCSetupRequest())) -> value v_ReceivedAsp;
             if(match(v_ReceivedAsp.Signalling.Rrc.Ccch.message_.c1.rrcSetupRequest.rrcSetupRequest.establishmentCause, mcs_PriorityAccess))
          {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 2");
          }
          else
          {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
          }

    //@siclog "Steps 3 - 6" siclog@
    //Steps 2 to 5 of the mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 12. (Note 1)
    //Note 1: The SS includes a 5GS network feature support IE in the REGISTRATION ACCEPT message configured as Table 11.3.6.3.3-3.
    f_NR_RRCSetup_Def(nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def));
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }

    f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell12, v_GMM_MobilityInfo, -, -, noRrcConnectionRelease, cs_NG_NetworkFeatureSupport_TC1136);

    //@siclog "Step 6A - 6D" siclog@
    //@sic R5-201227 sic@
    //Steps 5-8 as defined in TS 38.508-1 [4] Table Table 4.5.4.2-3.
    f_NR_RRC_ActivateSecurity (nr_Cell12, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
    f_NR_RRCReconfigExistingDRBs (nr_Cell12, omit);

	f_NR5GC_CloseUE_TestLoopModeB (nr_Cell12, '02'O); 
    //@siclog "Step 7" siclog@
    //The SS transmits one IP PDU.
    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    v_QoS_Id := f_NR_GetQosFlowIdentification_ForDRBId(v_DRBId);
    DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell12, v_QoS_Id, {v_IPData}));

    //@siclog "Step 8" siclog@
    //The SS transmits an RRCRelease message and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell12);

    //@siclog "Step 9" siclog@
    //Check: Does the UE transmit a RRCSetupRequest message including establishmentCause of mcs-PriorityAccess within 10s?
    t_WaitTime.start(v_TimerValue1);
    f_NR_RRC_SetupRequest_Common(nr_Cell12, t_WaitTime, mcs_PriorityAccess);  //@sic R5s201387 sic@

    //@siclog "Step 10" siclog@
    //SS transmit an RRCSetup message.
    f_NR_RRCSetup_Def(nr_Cell12);

    //@siclog "Step 11" siclog@
    //The UE transmits an RRCSetupComplete message to confirm the successful completion of the connection establishment.
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                               cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def),
                                               tsc_SHT_IntegrityProtected);
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), v_S_TMSI, '0101'B)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }

    //@siclog "Step 11A - 11D" siclog@
    //@sic R5-201227 sic@
    //Steps 5 to 8 of the NR RRC_CONNECTED procedure in TS 38.508-1 Table 4.5.4.2-3 are performed.    
	
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Step 12" siclog@
    //The UE loops back the IP PDU.
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell12, v_QoS_Id, {v_IPData}));

    //EXCEPTION: Steps 14a1-14a19 describe behaviour that depends on UE configuration; the "lower case letter" identifies a step sequence that takes place if inactiveState is configured
    //@sic R5-211409 sic@
    if(pc_inactiveState){

    //@siclog "Step 13" siclog@
    //The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
    f_NR_RRCRelease(nr_Cell12);	

      //@siclog "Step 14a1" siclog@
      //IF pc_inactiveState THEN the SS changes the SIB1 of NR Cell 1 to set the uac-BarringInfo and adjusts the NR Cells power levels according to row "T2" in table 11.3.6.3.2-1/2.
      f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                           cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(7,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));
      f_NR_ModifySysinfo(nr_Cell1, false); //Modify Sysinfo in SS
      
      f_NR_SetCellPowerList(v_CellPowerList_T2);
      
      //@siclog "Steps 14a2 - 14a6" siclog@
      //Steps 1 to 5 of the mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 are performed on NR Cell 1. (Note 2)
      //Note 2: The UE performs registration and the RRC connection is released.
      v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell1, tsc_NR_RRC_TI_Def, ?, tsc_SHT_IntegrityProtected); // @sic R5-200477 sic@
      if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility)) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
      }
      f_NR5GC_MobilityRegistration_Steps4_6(nr_Cell1,v_GMM_MobilityInfo, omit, omit, noRrcConnectionRelease);
      
      //@siclog "Step 14a7 - 14a10" siclog@
      //@sic R5-201227 sic@
      //Steps 5-8 as defined in TS 38.508-1 [4] Table Table 4.5.4.2-3.
      f_NR_RRC_ActivateSecurity (nr_Cell1, valueof(v_ReceivedMsg.SecurityProtection.NasCount));
      f_NR_RRCReconfigExistingDRBs (nr_Cell1, omit);
      
      //@siclog "Step 14a11" siclog@
      //The SS transmits one IP PDU.
      DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id, {v_IPData}));
      
      //@siclog "Step 14a12" siclog@
      //The SS transmits an RRCRelease message with suspend configuration and move the UE to RRC_INACTIVE.
      f_NR_RRC_InactiveDef(nr_Cell1);
      
      //@siclog "Step 14a13 siclog@
      //Check: Does the UE transmit a RRCResumeRequest message including resumeCause of mcs-PriorityAccess within 10s?
      //Note 4: The wait time 10s is selected to be less than T390 minimum = (0.7 + 0.6 * rand) * uac-BarringTime(16s) = 11.2s when 'rand' takes the minimum value of 0.
      t_WaitTime.start(v_TimerValue1);
      alt {
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mcs_PriorityAccess)))
        {
           t_WaitTime.stop;
           f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22");
        }
        [] t_WaitTime.timeout {}
      }
      
      //@siclog "Step 14a14 siclog@
      //SS changes SIB1 according to Table 11.3.6.3.3-5. And the SS notifies the UE of change of System Information on NR Cell 1 by send Short Message on PDCCH using P-RNTI.
      f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1, omit);
      f_NR_ModifySysinfo(nr_Cell1, true, RRC_INACTIVE);
  
	  f_NR_WaitModificationPeriods(nr_Cell1);  
	  f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);
	  SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mcs_PriorityAccess)));
  
	  //The SS transmits an RRCResume message.
	  v_DRBId := f_NR_GetDefaultDRB_ForFirstNonIMSPDUSession();
	  v_NR_RLC_BearerToAddModList[0] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_SRB(tsc_SRB2), omit, omit, cs_38508_LogicalChannelConfig_SRB(tsc_SRB2), true_);
	  v_NR_RLC_BearerToAddModList[1] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_DRB(v_DRBId), omit, omit, omit, true_);
	  v_NR_UplinkBWP.Dedicated.R15.pusch_Config := omit;
	  v_SpCellConfig := f_NR_GetSpCellConfigDef(nr_Cell1, cs_NR_ServingCellConfigDef(omit, 0, omit, cs_NR_UplinkConfigDef(v_NR_UplinkBWP.Dedicated.R15, omit)));
	  v_CellGroupConfig := cs_38508_CellGroupConfig(f_NR_CellInfo_Get_CellGroupId(nr_Cell1), v_NR_RLC_BearerToAddModList, -, -, -, v_SpCellConfig);
	  
	  v_RadioBearerConfig := cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod_ReestablishPDCP(tsc_SRB2)}, {cs_NR_DRB_ToAddMod(omit, v_DRBId, true_, -, omit)});
	  
	  f_NR_SendRRCResume_Def(nr_Cell1, -, -, v_CellGroupConfig, v_RadioBearerConfig);
	  
	  //@siclog "Step 5D" siclog@
	  //The UE transmits an RRCResumeComplete message.
	  f_NR_RRCResumeComplete_Def(nr_Cell1);
	  
	  //@siclog "Step 5E" siclog@
	  //The SS transmits one IP PDU.
      DRB.send(cas_NR_DRB_COMMON_REQ_SDAP_SDUList(nr_Cell1, v_QoS_Id, {v_IPData}));
    
	  //@siclog "Step 5F" siclog@
	  //The SS transmits an RRCRelease message with suspendConfig IE and move the UE to RRC_Inactive state.
	  f_NR_RRC_InactiveDef(nr_Cell1); 

	        
      //@siclog "Step 14a15 siclog@
      //Check: Does the UE transmit a RRCResumeRequest message including resumeCause of mcs-PriorityAccess within 30 s?(Note 3)
      //Note 3: The wait time 30s is selected to cover (0.7 + 0.6 * rand) * uac-BarringTime(16s) = 20.8s + 1s (IP PDU delay timer) + 5.12s (modification period) = 26.92s rounded up to 27s when 'rand' takes the maximum value of 1.
      t_WaitTime.start(v_TimerValue2);
      alt {
        [] SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mcs_PriorityAccess)))
        {
           t_WaitTime.stop;
           f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 24");
        }
        [] t_WaitTime.timeout {}
      }
      
      //@siclog "Step 14a16 siclog@
      //SS transmit an RRCResume message.
      f_NR_SendRRCResume_Def(nr_Cell1);
	  
      //@siclog "Step 14a17" siclog@
      //The UE transmits an RRCResumeComplete message to confirm the successful completion of the connection establishment.
      f_NR_RRCResumeComplete_Def(nr_Cell1);
      
      //@siclog "Step 14a18 siclog@
      //The UE loop back the IP PDU.
      DRB.receive(car_NR_DRB_COMMON_IND_SDAP_SDUList(nr_Cell1, v_QoS_Id, {v_IPData}));
      
      //@siclog "Step 14a19 siclog@
      //The SS transmits an RRCRelease message to release RRC connection and move the UE to RRC_IDLE.
      //REMOVE f_NR_RRCRelease(nr_Cell1);
    }

    //@siclog "Steps 15 - 28 siclog@
    //@sic R5-211409 sic@
    //Void
  }

